
% Dynamic Macroeconomics
% Chapter 3
%
% Ramsey Model in discrete time
% _____________________________________
%
% Two permanent shocks are introduced in order to study 
% the dynamic behavior of the model
%
% 1. A 1 per cent shock to the pure rate of time preference (endval 
x 1.01) 
% 2. A 1 per cent shock to total factor productivity (endval a 1.01)
% 
% _____________________________________
%
% Endogenous Variables: 
% y output, k capital, c consumption
% w real wage, r real interest rate, s savings rate
%
% Exogenous Variables and Shocks
% a total factor productivity, x shock to the pure rate of time 
preference
% z shock to population growth
%
% Parameters
% alpha share of capital in production, delta depreciation rate
% n population growth rate, g rate of technical progress
% rho pure rate of time preference
% theta inverse of elasticity of intertemporal substitution
%
% ______________________________________

var y k c w r s gy;
varexo a x z;

parameters alpha delta n g rho theta;

alpha=0.333;
delta=0.03;
n=0.01;
g=0.02;
rho=0.02;
theta=1;

model;

y=a*(k(-1)^alpha);
r=(alpha*a*k^(alpha-1))-delta;
c(+1)=((((1+r)/(1+(rho*x)))^(1/theta))*(1/(1+g)))*c;
k=(1/((1+n*z)*(1+g)))*((y-c)+(1-delta)*k(-1));
w=(1-alpha)*a*k(-1)^(alpha);
s=(y-c)/y;
gy=(y-y(-1))/y(-1);



end;

initval;

k=5.3;
c=1.38;
y=1.7;
a=1;
r=0.075;
w=1.16;
x=1;
z=1;

end;

steady;

endval;

k=5.3;
c=1.38;
y=1.7;
a=1.01;
r=0.075;
w=1.16;
x=1.0;
z=1;

end;

steady;
check;

simul(periods=100);

% Plotting Capital Output Consumption Real Interest Rate Real Wage 
Savings
% Rate

subplot(3,2,1); plot(k(1:100,1)); title('Capital');
subplot(3,2,2); plot(y(1:100,1)); title('Output');
subplot(3,2,3); plot(c(1:100,1)); title('Consumption');
subplot(3,2,4); plot(w(1:100,1)); title('Real Wage');
subplot(3,2,5); plot(r(1:100,1)); title('Real Interest Rate');
subplot(3,2,6); plot(s(1:100,1)); title('Savings Rate');


